Chapter 1
Introduction

1.1 Introduction to Maple

1.1.1 Getting Started with Maple

Some Maple basics are presented in this chapter as a convenience for the reader. Two Maple books[1, 2] that have proven to be useful are given as references 1 and 2 at the end of this chapter. Maple can be started either from the shortcut on the desktop or from Start → Programs → Maple 12. This opens a new Maple worksheet in the Maple environment. You should usually type 'restart' as the first command in your Maple worksheets.

```
> restart;
```

This restart command clears all the stored variables and restarts the worksheet every time it is executed.

Numerical values can be assigned to variables in Maple by using the characters ':= after x, for example. That is, to assign the value 2 to the variable x, the colon and equal sign ':=' characters are used together. You can use the # sign to add comments

```
> x:=2; # an assignment statement.
```

$$x := 2$$

Note that ':=' is the assignment operator which assigns an expression or number (2) to a variable named x. If the colon is not used, the value is not assigned. For example, 2 is not assigned to y by using '=' only. For example, type

```
> y=2;
```

$$y = 2$$

Now type both x and y to see their values.

```
> x;
```

$$2$$

```
> y;
```

$$y$$

This shows that ‘:=’ assigned the value 2 to x whereas ‘=’ did not assign 2 to y.

One can use Maple to do numerical and symbolic calculations. A few examples are shown next.

```
> x^2;
```

$$4$$

```
> x^2.;
```

$$4.$$

```
> sqrt(x);
```

$$\sqrt{2}$$

```
> x^0.5;
```

$$1.414213562$$

```
> abs(x);
```

$$2$$

```
> -x;
```

$$-2$$

```
> x+y;
```

$$-2+y$$

```
> abs(-2);
```

$$2$$

The imaginary number $\sqrt{-1}$ is designated as I in Maple:

```
> (-1)^(1/2);
```

$$I$$

The Maple command ‘evalf’ provides numeric evaluation and the ‘eval’ command yields a symbolic evaluation:

```
> evalf(sqrt(2));
```

$$1.414213562$$

```
> eval(sqrt(2));
```

$$\sqrt{2}$$

Symbolic variables can also be assigned to names as follows:

```
> z:=y;
```

$$z := y$$

```
> z;
```

$$y$$

Differentiation can be done by using the ‘diff’ command:

```
> diff(y,y);
```

$$1$$

```
> diff(y^2,y);
```

$$2\,y$$

Integration can be done by using the ‘int’ command:

```
> int(y,y);
```

$$\frac{y^2}{2}$$

Maple can also do definite integration:

```
> int(y,y=0..1);
```

$$\frac{1}{2}$$

1.1.2 Plotting with Maple

Plots can be made in Maple using the ‘plot’ command:

```
> plot(y,y=0..1);
```



Fig. 1.1 Maple plot of y = y

```
> plot(y^2,y=0..1);
```



Fig. 1.2 Maple plot of $y^2 = y$

To plot both curves on the same graph in a box use the following command.

```
> plot([y,y^2],y=0..1,axes=boxed);
```



Fig. 1.3 Maple plot of y and y^2 vs y

1.1.3 Solving Linear and Nonlinear Equations

One can solve equations in Maple using the 'solve' and 'fsolve' commands. The 'solve' command is used to solve linear equations in symbolic form and the 'fsolve' command is used to solve linear and nonlinear equations numerically. For example,

```
> restart:
```

```
> eq:=x+2;
```

$$eq := x + 2$$

```
> solve(eq);
```

$$-2$$

Maple can solve equations in symbolic form also:

```
> eq:=x-a;
```

$$eq := x - a$$

```
> solve(eq);
```

$$\{ a = x, x = x \}$$

This solution says that either x = x or a = x. To solve specifically for x

```
> solve(eq,x);
```

$$a$$

Note that a has not been assigned to x which can by seen by typing x:

```
> x;
```

$$x$$

One can assign the value of a to x by solving the above equation for x:

```
> eq:=x-a;
```

$$eq := x - a$$

```
> x:=solve(eq,x);
```

$$x := a$$

One can use the 'fsolve' command in Maple to solve equations numerically:

```
> eq1:=y+1;
```

$$eq1 := y + 1$$

```
> fsolve(eq1,y);
```

$$-1.$$

Note that 'fsolve' returns a floating point number with a decimal point.

Two or more nonlinear equations can be solved by using ‘fsolve’. For example, consider finding the solutions (x and y) for the following two equations.

```
> restart:
> eq1:=x+tan(y)=1;
```

$$eq1 := x + \tan(y) = 1$$

```
> eq2:=y^2+tan(x)=1;
```

$$eq2 := y^2 + \tan(x) = 1$$

```
> fsolve({eq1,eq2},{x,y});
```

$$\{ x = -3.858064894, y = 1.367788596 \}$$

One can find other solutions to these equations by restricting the ranges of x and y:

```
> fsolve(f#,{x=1..3,y=1..3});
```

$$\{ x = 1.760535729, y = 2.491382707 \}$$

1.1.4 Matrix Operations

Maple has a package for solving linear algebra problems which can be called by using the ‘with(linalg)’ command.

```
> restart:
> with(linalg):
```

Warning, the protected names norm and trace have been redefined and unprotected.

Maple is capable of doing a variety of matrix operations. For example, let A and B be 2 x 2 matrices which can be entered as follows:

```
> A:=matrix(2,2,[1,2,3,4]);
```

$$A := \begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix}$$

```
> B:=matrix(2,2,[1,1,3,2]);
```

$$B := \begin{bmatrix} 1 & 1 \\ 3 & 2 \end{bmatrix}$$

Use the ‘evalm’ command to perform matrix operations. For example, matrix addition and subtraction can be done:

```
> evalm(A+B);
```

$$\begin{bmatrix} 2 & 3 \\ 6 & 6 \end{bmatrix}$$

```
> evalm(A-B);
```

$$\begin{bmatrix} 0 & 1 \\ 0 & 2 \end{bmatrix}$$

Multiplication of matrices requires using evalm and '&*':

```
> evalm(A&*B);
```

$$\begin{bmatrix} 7 & 5 \\ 15 & 11 \end{bmatrix}$$

The determinant of a matrix can be found by using 'det':

```
> det(A);
```

$$-2$$

and

```
> det(B);
```

$$-1$$

Matrices can be inverted by using the 'inverse command':

```
> inverse(A);
```

$$\begin{bmatrix} -2 & 1 \\ \frac{3}{2} & \frac{-1}{2} \end{bmatrix}$$

```
> inverse(inverse(A));
```

$$\begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix}$$

The transpose of a matrix can be obtained also

```
> transpose(A);
```

$$\begin{bmatrix} 1 & 3 \\ 2 & 4 \end{bmatrix}$$

A particular element of a matrix can be printed easily:

```
> A[1,1];
```

$$1$$

A matrix can be raised to a power by using the ‘evalm’ command:

```
> evalm(A^2);
```

$$\begin{bmatrix} 7 & 10 \\ 15 & 22 \end{bmatrix}$$

The characteristic polynomial, eigenvalues, and eigenvectors of a matrix can be obtained as follows:

```
> charpoly(A,lambda);
```

$$\lambda^2 - 5\,\lambda - 2$$

```
> eigenvalues(A);
```

$$\frac{5}{2} + \frac{\sqrt{33}}{2},\ \frac{5}{2} - \frac{\sqrt{33}}{2}$$

```
> eigenvectors(A);
```

$$\left[\frac{5}{2} + \frac{\sqrt{33}}{2},\ 1,\ \left\{\left[1,\ \frac{3}{4} + \frac{\sqrt{33}}{4}\right]\right\}\right], \left[\frac{5}{2} - \frac{\sqrt{33}}{2},\ 1,\ \left\{\left[1,\ \frac{3}{4} - \frac{\sqrt{33}}{4}\right]\right\}\right]$$

or

```
> eigenvects(A);
```

$$\left[\frac{5}{2} + \frac{\sqrt{33}}{2},\ 1,\ \left\{\left[1,\ \frac{3}{4} + \frac{\sqrt{33}}{4}\right]\right\}\right], \left[\frac{5}{2} - \frac{\sqrt{33}}{2},\ 1,\ \left\{\left[1,\ \frac{3}{4} - \frac{\sqrt{33}}{4}\right]\right\}\right]$$

Matrices can be raised to various powers and added. For example, let

```
> eq:=A+A^2+A^3;
```

$$eq := A + A^2 + A^3$$

```
> evalm(eq);
```

$$\begin{bmatrix} 45 & 66 \\ 99 & 144 \end{bmatrix}$$

Maple’s ‘Id’ command can be used to generate an identity matrix:

```
> Id:=band([1],2);
```

$$Id := \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

Elements of a matrix can be in symbolic form and a variety of matrix operations can be performed:

```
> A:=matrix(2,2,[a,b,c,d]);
```

$$A := \begin{bmatrix} a & b \\ c & d \end{bmatrix}$$

```
> transpose(A);
```

$$\begin{bmatrix} a & c \\ b & d \end{bmatrix}$$

```
> inverse(A);
```

$$\begin{bmatrix} \frac{d}{a\,d-b\,c} & -\frac{b}{a\,d-b\,c} \\ -\frac{c}{a\,d-b\,c} & \frac{a}{a\,d-b\,c} \end{bmatrix}$$

```
> evalm(A&*B);
```

$$\begin{bmatrix} a+3\,b & a+2\,b \\ c+3\,d & c+2\,d \end{bmatrix}$$

A matrix can be multiplied with a scalar:

```
> evalm(2*A);
```

$$\begin{bmatrix} 2\,a & 2\,b \\ 2\,c & 2\,d \end{bmatrix}$$

Eigenvalues can be obtained:

```
> eigenvalues(A);
```

$$\frac{a}{2}+\frac{d}{2}+\frac{\sqrt{a^2-2\,a\,d+d^2+4\,b\,c}}{2},\ \frac{a}{2}+\frac{d}{2}-\frac{\sqrt{a^2-2\,a\,d+d^2+4\,b\,c}}{2}$$

Eigenvectors can be obtained:

```
> eigenvects(A);
```

$$\left[\frac{a}{2}+\frac{d}{2}+\frac{\sqrt{a^2-2\,a\,d+d^2+4\,b\,c}}{2}, 1, \left\{\left[-\frac{-\frac{a}{2}+\frac{d}{2}-\frac{\sqrt{a^2-2\,a\,d+d^2+4\,b\,c}}{2}}{c}, 1\right]\right\}\right], \left[\frac{a}{2}+\frac{d}{2}-\frac{\sqrt{a^2-2\,a\,d+d^2+4\,b\,c}}{2}, 1, \left\{\left[-\frac{-\frac{a}{2}+\frac{d}{2}+\frac{\sqrt{a^2-2\,a\,d+d^2+4\,b\,c}}{2}}{c}, 1\right]\right\}\right]$$

The exponential matrix of a matrix can be obtained as follows:

```
> exponential(B,t);
```

$$\left[\frac{1}{2}\mathrm{e}^{\left(\frac{t\,(3+\sqrt{13}\,)}{2}\right)}+\frac{1}{26}\sqrt{13}\,\mathrm{e}^{\left(-\frac{t\,(-3+\sqrt{13}\,)}{2}\right)}-\frac{1}{26}\sqrt{13}\,\mathrm{e}^{\left(\frac{t\,(3+\sqrt{13}\,)}{2}\right)}+\frac{1}{2}\mathrm{e}^{\left(-\frac{t\,(-3+\sqrt{13}\,)}{2}\right)},\right.$$
$$\left.-\frac{1}{13}\sqrt{13}\,\mathrm{e}^{\left(-\frac{t\,(-3+\sqrt{13}\,)}{2}\right)}+\frac{1}{13}\sqrt{13}\,\mathrm{e}^{\left(\frac{t\,(3+\sqrt{13}\,)}{2}\right)}\right]$$
$$\left[-\frac{3}{13}\sqrt{13}\,\mathrm{e}^{\left(-\frac{t\,(-3+\sqrt{13}\,)}{2}\right)}+\frac{3}{13}\sqrt{13}\,\mathrm{e}^{\left(\frac{t\,(3+\sqrt{13}\,)}{2}\right)},\right.$$
$$\left.\frac{1}{2}\mathrm{e}^{\left(\frac{t\,(3+\sqrt{13}\,)}{2}\right)}-\frac{1}{26}\sqrt{13}\,\mathrm{e}^{\left(-\frac{t\,(-3+\sqrt{13}\,)}{2}\right)}+\frac{1}{26}\sqrt{13}\,\mathrm{e}^{\left(\frac{t\,(3+\sqrt{13}\,)}{2}\right)}+\frac{1}{2}\mathrm{e}^{\left(-\frac{t\,(-3+\sqrt{13}\,)}{2}\right)}\right]$$

The ‘map’ command can be used to differentiate and integrate each element in a matrix:

```
> A:=matrix(2,2,[x,a*x,1/x,c]);
```

$$A := \begin{bmatrix} x & a\,x \\ \frac{1}{x} & c \end{bmatrix}$$

```
> map(diff,A,x);
```

$$\begin{bmatrix} 1 & a \\ -\frac{1}{x^2} & 0 \end{bmatrix}$$

```
> map(int,A,x);
```

$$\begin{bmatrix} \frac{x^2}{2} & \frac{a\,x^2}{2} \\ \ln(x) & c\,x \end{bmatrix}$$

```
> map(int,A,x=0..1);
```

$$\begin{bmatrix} \frac{1}{2} & \frac{a}{2} \\ \infty & c \end{bmatrix}$$

1.1.5 *Differential Equations*

Maple's 'dsolve' command can be used to obtain analytical and series solutions for differential equations. Differential equations are discussed in more detail in chapters 2 and 3. In this section, some Maple commands are introduced to solve relatively simple differential equations.

```
> restart:
```

You have to use y(x) if you are trying to solve y as a function of x (y is the dependent variable and x is the independent variable)

```
> eq:=diff(y(x),x)=x;
```

$$eq := \frac{d}{dx}\,\mathrm{y}(x) = x$$

```
> dsolve(eq,y(x));
```

$$\mathrm{y}(x) = \frac{x^2}{2} + \_C1$$

Note that the constant _C1 is returned as part of the solution. If you specify the initial condition, Maple can be used to obtain the complete solution:

```
> dsolve({eq,y(0)=1},y(x));
```

$$\mathrm{y}(x) = \frac{x^2}{2} + 1$$

Second order equations can also be solved with 'dsolve':

```
> eq:=y(x)+diff(y(x),x$2)=x^3;
```

$$eq := \mathrm{y}(x) + \left(\frac{d^2}{dx^2}\,\mathrm{y}(x)\right) = x^3$$

```
> dsolve(eq,y(x));
```

$$\mathrm{y}(x) = \sin(x)\,\_C2 + \cos(x)\,\_C1 + x\,(-6 + x^2)$$

Note that there are two constants, _C2 and _C1, in this case. The D(y)(x) command can be used to set the derivate of y as an initial condition at x=0 $\left(\frac{dy}{dx} = 0, \text{ e.g.}\right)$, and the other initial condition $\left(y(0) = 1\right)$ can be set easily also:

```
> dsolve({eq,y(0)=1,D(y)(0)=0},y(x));
```

$$\mathrm{y}(x) = 6\sin(x) + \cos(x) + x\,(-6 + x^2)$$

Next, store the right hand side (rhs) in ya and then plot ya:

```
> ya:=rhs(dsolve({eq,y(0)=1,D(y)(0)=0},y(x)));
```

$$ya := 6\sin(x) + \cos(x) + x\,(-6 + x^2)$$

```
> plot(ya,x=0..1);
```



Fig. 1.4 Maple plot of ya vs x

Maple's 'dsolve' can be used to solve nonlinear equations. For example, consider the following equation:

```
> eq:=diff(y(x),x$2)=y(x)^2;
```

$$eq := \frac{d^2}{dx^2}\,\mathrm{y}(x) = \mathrm{y}(x)^2$$

Solve this equation by using 'dsolve':

```
> dsolve(eq,y(x));
```

$$\int^{\mathrm{y}(x)} \frac{3}{\sqrt{6\,\_a^3 - 3\,\_C1}}\,d\_a - x - \_C2 = 0, \int^{\mathrm{y}(x)} -\frac{3}{\sqrt{6\,\_a^3 - 3\,\_C1}}\,d\_a - x - \_C2 = 0$$

Maple gives the solution as an integral. Instead one can get a series solution by specifying 'type = series' in 'dsolve' as follows:

```
> dsolve(eq,y(x),type =series);
```

$$y(x) = y(0) + D(y)(0)\,x + \frac{1}{2}\,y(0)^2\,x^2 + \frac{1}{3}\,y(0)\,D(y)(0)\,x^3 + \left(\frac{1}{12}\,y(0)^3 + \frac{1}{12}\,D(y)(0)^2\right) x^4 + \frac{1}{12}\,y(0)^2\,D(y)(0)\,x^5 + O(x^6)$$

Consider another nonlinear differential equation.

```
> eq:=diff(y(x),x)=-tan(x)+exp(-y(x));
```

$$eq := \frac{d}{dx}\,y(x) = -\tan(x) + e^{(-y(x))}$$

```
> dsolve(eq,y(x));
```

$$y(x) = -\ln\left(\frac{1}{\cos(x)\,(\_C1 + \ln(\sec(x) + \tan(x)))}\right)$$

Use the type = series option to obtain a series solution.

```
> ya:=rhs(dsolve({eq,y(0)=1},y(x),type=series));
```

$$ya := 1 + e^{(-1)}\,x + \left(-\frac{1}{2}\,(e^{(-1)})^2 - \frac{1}{2}\right) x^2 + \frac{1}{3}\left((e^{(-1)})^2 + \frac{1}{2}\right) e^{(-1)}\,x^3 + \left(-\frac{1}{12} - \frac{1}{4}\,(e^{(-1)})^4 - \frac{1}{6}\,(e^{(-1)})^2\right) x^4 + \left(\frac{1}{24}\,e^{(-1)} + \frac{1}{5}\,(e^{(-1)})^5 + \frac{1}{6}\,(e^{(-1)})^3\right) x^5 + O(x^6)$$

```
> ya:=evalf(ya);
```

$$ya := 1. + 0.3678794412\,x - 0.5676676416\,x^2 + 0.07790892966\,x^3 - 0.1104681236\,x^4 + 0.02497374418\,x^5 + O(x^6)$$

One can remove the order term $\left(0\left(x^6\right)\right)$ in the series by using the 'convert' command:

```
> ya:=convert(ya,polynom);
```

$$ya := 1. + 0.3678794412\,x - 0.5676676416\,x^2 + 0.07790892966\,x^3 - 0.1104681236\,x^4 + 0.02497374418\,x^5$$

```
> plot(ya,x=0..1);
```



Fig. 1.5 Maple plot of ya vs x

One can also use 'dsolve' to solve boundary value problems. Consider heat transfer in a fin:[3]

```
> eq:=diff(y(x),x$2)=H^2*y(x);
```

$$eq := \frac{d^2}{dx^2}\mathrm{y}(x) = H^2\,\mathrm{y}(x)$$

where H is a parameter. The governing equation can be solved without specifying the boundary conditions as:

```
> dsolve(eq,y(x));
```

$$\mathrm{y}(x) = \_C1\,\mathrm{e}^{(H\,x)} + \_C2\,\mathrm{e}^{(-H\,x)}$$

Suppose the boundary conditions are at x=0, y=1, and at x=1, $\dfrac{dy}{dx} = 0$. If one of the boundary conditions is specified, Maple gives a solution with one constant.

```
> ya:=rhs(dsolve({eq,y(0)=1},y(x)));
```

$$ya := (-\_C2 + 1)\,\mathrm{e}^{(H\,x)} + \_C2\,\mathrm{e}^{(-H\,x)}$$

The constant _C2 can be obtained by using the boundary condition at x = 1:

```
> diff(ya,x);
```

$$(-\_C2+1)\,H\,e^{(H\,x)}-\_C2\,H\,e^{(-H\,x)}$$

```
> bc:=subs(x=1,diff(ya,x));
```

$$bc := (-\_C2+1)\,H\,e^{H}-\_C2\,H\,e^{(-H)}$$

```
> _C2:=solve(bc,_C2);
```

$$\_C2 := \frac{e^{H}}{e^{H}+e^{(-H)}}$$

The complete solution is obtained by using Maple's 'simplify' command as follows:

```
> ya:=simplify(ya);
```

$$ya := \frac{e^{(-H+H\,x)}+e^{(-H\,(-1+x))}}{e^{H}+e^{(-H)}}$$

Plot the solution ya with H=1:

```
> plot(subs(H=1,ya),x=0..1);
```



Fig. 1.6 Maple plot of ya vs x for H=1

Next, plot the solution ya with H=3 and use points instead of a line.

```
> plot(subs(H=3,ya),x=0..1,style=point);
```



Fig. 1.7 Maple point plot of ya vs x for H=3

1.1.6 Laplace Transformations

Maple can be used to obtain Laplace transforms and inverse Laplace transforms of functions symbolically. For this purpose, the package 'with(inttrans)' is used:

```
> restart:
> with(inttrans):
```

Suppose we want to find the Laplace transform of t, we use

```
> f(t):=t;
```

$$f(t) := t$$

```
> laplace(f(t),t,s);
```

$$\frac{1}{s^2}$$

Laplace transforms for different functions can be obtained easily:

```
> laplace(f(t)*t,t,s);
```

$$\frac{2}{s^3}$$

```
> f(t):=exp(-t);
```

$$f(t) := e^{(-t)}$$

```
> laplace(f(t),t,s);
```

$$\frac{1}{1+s}$$

Both Laplace and inverse Laplace transforms can be obtained with Maple.

```
> f(t):=sin(t);
```

$$f(t) := \sin(t)$$

```
> f(s):=laplace(f(t),t,s);
```

$$f(s) := \frac{1}{s^2+1}$$

```
> invlaplace(f(s),s,t);
```

$$\sin(t)$$

Inverse Laplace transforms for different functions can be also obtained:

```
> f(s):=1/sqrt(s);
```

$$f(s) := \frac{1}{\sqrt{s}}$$

```
> invlaplace(f(s),s,t);
```

$$\frac{1}{\sqrt{\pi\, t}}$$

```
> f(s):=1/(s)^(3/2);
```

$$f(s) := \frac{1}{s^{(3/2)}}$$

```
> invlaplace(f(s),s,t);
```

$$\frac{2\sqrt{t}}{\sqrt{\pi}}$$

```
> f(s):=exp(-sqrt(s));
```

$$f(s) := e^{(-\sqrt{s})}$$

```
> invlaplace(f(s),s,t);
```

$$\frac{1}{2}\,\frac{e^{\left(-\frac{1}{4\,t}\right)}}{\sqrt{\pi}\, t^{(3/2)}}$$

Unfortunately, Maple cannot find the inverse Laplace transform for complicated functions:

```
> f(s):=1/sinh(sqrt(s));
```

$$f(s) := \frac{1}{\sinh(\sqrt{s})}$$

```
> invlaplace(f(s),s,t);
```

$$\operatorname{invlaplace}\left(\frac{1}{\sinh(\sqrt{s})}, s, t\right)$$

This does not mean that the inverse Laplace transform does not exist; instead, one has to use advanced techniques for finding the desired inverse Laplace transform (see chapter 8 for details).

1.1.7 Do Loop

It is possible to carry out a sequence of steps using a 'do loop.' The syntax is

> **for** *variables* **in** *expresion* **do** *statement sequence* **end do**;

statement sequence

We can prepare a set of differential equations by using a 'do loop'. Use the shift enter keys to add the extra line after the "do" as shown in the following worksheet.

```
> restart:
> N:=3;
```

$$N := 3$$

```
> for i from 1 to N do  #Use the shift enter keys to add each new line in a do loop.
diff(y[i](t),t)=(y[i+1](t),-y[i-1](t));
od;
```

$$\frac{d}{dt} y_1(t) = (y_2(t), -y_0(t))$$
$$\frac{d}{dt} y_2(t) = (y_3(t), -y_1(t))$$
$$\frac{d}{dt} y_3(t) = (y_4(t), -y_2(t))$$

1.1.8 *While Loop*

It is possible to carry out a sequence of statements or commands until a prescribed condition is satisfied. The 'while' command can be used to do this. The general statement is of the form:

> **while** *conditions* **do** *statement sequence* **end do**;

For example, one could use the following worksheet to determine the square of a number.

> *restart* :

> $N := 1;$

$$N := 1$$

> **while** $N < 3$ **do** $A := N^2;\ N := N + 1;$ **end do**;

$$A := 1$$

$$N := 2$$

$$A := 4$$

$$N := 3$$

>

1.1.9 *Write Data Out Example*

Data can be generated and written out to a text file (i.e., a . txt file). For example, we can use Maple to solve the second order ordinary differential equation

$$\frac{d^2u}{dx^2} = u \tag{1.1}$$

with the following boundary conditions:

$$u(0) = 0.21 \tag{1.2}$$

and

$$\left.\frac{du}{dx}\right|_{x=1} = 0 \tag{1.3}$$

The result is

$$u(x) = 0.21\frac{\cosh(x-1)}{\cosh(1)} \tag{1.4}$$

Values for this analytical solution at various values of x can be generated and exported to a text file as shown in the worksheet below.

> *restart:with(plots):with(linalg):*

Specify the values for x at which the analytical solution will be calculated and later exported. Make sure you have the whole range (i.e., 0 to 1) of x included.

> $xdata := matrix\left(11, 1, \left[seq\left(\frac{k}{10.0}, k = 0..10\right)\right]\right);$

$$xdata := \begin{bmatrix} 0. \\ 0.1000000000 \\ 0.2000000000 \\ 0.3000000000 \\ 0.4000000000 \\ 0.5000000000 \\ 0.6000000000 \\ 0.7000000000 \\ 0.8000000000 \\ 0.9000000000 \\ 1.000000000 \end{bmatrix}$$

Input the Governing Equation and Boundary Conditions:

> $eq := diff(u(x), x\$2) = u(x);$

$$eq := \frac{d^2}{dx^2} u(x) = u(x)$$

> $bcs := u(0) = 0.21, (D(u))(1) = 0;$

$$bcs := u(0) = 0.21, D(u)(1) = 0$$

Solve the differential equation and rearrange the results to the desired form:

> $u := rhs(dsolve(\{eq, bcs\}));$

$$u := \frac{21}{100} \frac{e\, e^{-x}}{e^{-1} + e} + \frac{21}{100} \frac{e^{-1} e^{x}}{e^{-1} + e}$$

> $u := convert(u, trig);$

$$u := \frac{21}{200} \frac{(\cosh(1) + \sinh(1))(\cosh(x) - \sinh(x))}{\cosh(1)} + \frac{21}{200} \frac{(\cosh(1) - \sinh(1))(\cosh(x) + \sinh(x))}{\cosh(1)}$$

> *simplify*(%);

$$\frac{21}{100}\,\frac{\cosh(1)\cosh(x)-\sinh(1)\sinh(x)}{\cosh(1)}$$

> *u* := *combine*(%);

$$u := \frac{21}{100}\,\frac{\cosh(-1+x)}{\cosh(1)}$$

Make sure the solution satisfies the boundary conditions:

> *evalf*(*subs*(*x* = 0, *u*));

$$0.210000000($$

> *evalf*(*subs*(*x* = 1, (D(*u*))(1)));

$$0.$$

Plot the results:

> *plot*(*u*, *x* = 0 ..1);



Fig. 1.8

Save the analytical solution at the x values specified at the beginning of this worksheet as uana[i].

\> **for** i **to** $rowdim(xdata)$ **do**
$uana_i := evalf\left(subs\left(x = xdata_{i,1}, u\right)\right);$

end do;

$$uana_1 := 0.2100000000$$
$$uana_2 := 0.195030728$$
$$uana_3 := 0.182013390$$
$$uana_4 := 0.170817704$$
$$uana_5 := 0.161331618$$
$$uana_6 := 0.153460193$$
$$uana_7 := 0.147124649$$
$$uana_8 := 0.142261579$$
$$uana_9 := 0.138822310$$
$$uana_{10} := 0.136772421$$
$$uana_{11} := 0.136091397$$

Export the Analytical Solution from Maple into the Text File:

\> $uanalytical := evalm(matrix(rowdim(xdata), 2, [seq([xdata[kd, 1], uana[kd]], kd = 1..rowdim(xdata))]));$

$$uanalytical := \begin{bmatrix} 0. & 0.2100000000 \\ 0.1000000000 & 0.1950307288 \\ 0.2000000000 & 0.1820133908 \\ 0.3000000000 & 0.1708177041 \\ 0.4000000000 & 0.1613316181 \\ 0.5000000000 & 0.1534601934 \\ 0.6000000000 & 0.1471246498 \\ 0.7000000000 & 0.1422615792 \\ 0.8000000000 & 0.1388223103 \\ 0.9000000000 & 0.1367724217 \\ 1.000000000 & 0.1360913975 \end{bmatrix}$$

Note: To export to a text file, the formats compatible with Maple are Arrays, Matrices, etc. For additional help please type ? writedata

> $fd := fopen("maple\_output.txt", WRITE, TEXT):$
$writedata(fd, uanalytical);$
$close(fd):$

Now, the output from the analytical solution is stored into a file called "maple_output.txt" in the folder where you saved this original Maple file.

1.1.10 Reading in Data from a Text File

Data can be read into Maple from a text file as shown below.
This worksheet is entitled ReadDataInExp.mws.

```
> restart:
```

Read in data from a text file named "maple_output.txt" located on the D drive under the folder named "ECHE700Sp09".

```
> fd := fopen("D:\\ECHE700Sp09\\maple_output.txt",READ);
data:=readdata(fd,2);
fclose(fd);
```

$$fd := 0$$

$data := [[0., 0.21], [0.1, 0.1950307288], [0.2, 0.1820133908], [0.3, 0.1708177041], [0.4, 0.1613316181], [0.5, 0.1534601934], [0.6, 0.1471246498], [0.7, 0.1422615792], [0.8, 0.1388223103], [0.9, 0.1367724217], [1., 0.1360913975]]$

Print the data.

```
> data:=evalm(data);
```

$$data := \begin{bmatrix} 0. & 0.21 \\ 0.1 & 0.1950307288 \\ 0.2 & 0.1820133908 \\ 0.3 & 0.1708177041 \\ 0.4 & 0.1613316181 \\ 0.5 & 0.1534601934 \\ 0.6 & 0.1471246498 \\ 0.7 & 0.1422615792 \\ 0.8 & 0.1388223103 \\ 0.9 & 0.1367724217 \\ 1. & 0.1360913975 \end{bmatrix}$$

1.1.11 Summary

In this chapter, some useful Maple commands were introduced. In section 1.1.1, basic Maple commands for assignment, evaluation, differentiation and integration were introduced. In section 1.1.2, commands for plotting were introduced. In section 1.1.3, linear and nonlinear equations were solved using Maple. Linear equations were solved symbolically (exactly) and nonlinear equations were solved numerically. In section 1.1.4, Maple's matrix operations such as addition, subtraction, finding the inverse, eigenvalues, etc. were introduced. In section 1.1.5, simple linear differential equations were solved using Maple's 'dsolve' command to obtain a closed form analytical solution. In addition, series solutions were obtained for certain nonlinear differential equations. In section 1.1.6, Laplace and inverse Laplace transforms for simple functions were obtained using Maple. In section 1.1.7, using a 'do loop' to carry out a sequence of steps using Maple was explained. In section 1.1.8, using a 'while loop' to carry out a sequence of statements or commands was explained using Maple. In section 1.1.9, steps for writing out data from Maple into a text file was discussed. In section 1.1.10, reading data into Maple from a text file was explained.

1.1.12 Problems

Create a different Maple worksheet for each of the following problems. Start each worksheet with the restart command.

1. Assign x = 4 and obtain the following results using Maple:

2. (1) x^2 (2) $1 + y/x$ (3) $\sqrt{x}$ (4) $\dfrac{1+y}{x}$

   (5) $1.2^x + x^2 - x^{1.2}$

3. Assign x = 2 and y = 3 and obtain results for the following using Maple:

   (1) sin(x) (b) arcsin(x) (i.e., $\sin^{-1}(x)$)(3) log(x) (4) log(y/x)

   (5) exp(x) (6) exp(x) + exp(y) – exp(xy) (7) log(y-x) + log(x-y)

4. Use Maple to find the derivatives of the following functions:

   (1) $x^2 - x\sin(x)$ (2) $\left(\dfrac{1}{1+x+x^2}\right)\log(x)$

5. Plot $\exp(-x^2)$ from x = 0 to 5. Use Maple to find the definite integral

   $$\int_0^L \exp(-x^2)dx$$

   for L = 0.1, 0.5, 1, and 2.

6. Use Maple to plot the following functions for x varying from 0 to 1:

   (1) $\exp(x)$ (2) $\exp(-x)$ (3) $1 - x + x^2$ (4) $x\,(1-x)$ (5) $x^2 - \log(x)$

7. Use Maple to plot the following functions for x varying from 0 to 1:

   (1) $\sin(\pi x)$ (2) $\cos(\pi x)$ (3) $\arcsin(x)$ (4) $\sin\left(\frac{\pi}{2}x\right)\exp(-x)$

8. Use Maple to solve the following equations symbolically (use the 'solve' command).

   (1) $ax^2 + bx + c = 0$

   (2) $x^3 - 1 = 0$

   (3) $x^4 - x^2 = 0$

   (4) $x + y = 3;\ x - y = 2$

   (5) $x + y = a;\ x - y = a - b$

   (6) $x + y + z = a;\ 2x + 3y + 4z = a + b + c;\ x - y - z = b$

   (7) $x + y + z = 6;\ xyz = 6;\ xy + yz + zx = 11$

9. Use Maple to solve the following equations numerically by using 'fsolve'. Find all the possible roots.

   (1) $x^3 - \tan(y) = xy;\ y^3 - \tan(x) = 1$

   (2) $x^2 + y^2 = 1;\ x^2 - y^2 = ¼$

   (3) $x^3 - 2x - \frac{1}{x} = 0$

10. Define the following matrices $A = \begin{bmatrix} 1 & 2 & 3 \\ 2 & 3 & 4 \\ 3 & 4 & 6 \end{bmatrix};\ B = \begin{bmatrix} 1 & 2 & 0 \\ 0 & 1 & 2 \\ 0 & 0 & 1 \end{bmatrix}$ in Maple.

    (1) Find A+B, A-B, AB and BA.

    (2) Find the determinant of A, B and AB.

    (3) Find A^{-1}, B^{-1}, A/B and B/A.

    (4) Find the eigenvalues and eigenvectors of A, B, AB and BA.

    (5) Find A^3, A + B + AB-BA.

11. Consider the matrix $A = \begin{bmatrix} \alpha & 1 & 0 \\ 0 & 1 & 1 \\ 0 & -1 & \alpha \end{bmatrix}$.

(1) Substitute $\alpha = 4$ in A and find the determinant, characteristic polynomial, eigenvalues, eigenvectors and inverse of A using Maple.
(2) Substitute $\alpha = 2$ in A and find the determinant, characteristic polynomial, eigenvalues, eigenvectors and inverse of A using Maple.
(3) Substitute $\alpha = 1$ in A and find the determinant, characteristic polynomial, eigenvalues, eigenvectors and inverse of A using Maple.
(4) Substitute $\alpha = 3$ in A and find the determinant, characteristic polynomial, eigenvalues, eigenvectors and inverse of A using Maple.

12. Consider the differential equation $\frac{dy}{dx} = -yx$; $y(0) = 1$. Use Maple to solve this differential equation by using the 'dsolve' command to obtain a closed-form solution. Obtain a series solution for the same. Plot the profiles.

13. Consider diffusion with a first order reaction in a rectangular catalyst pellet.[4] The governing equation in dimensionless form is $\frac{d^2y}{dx^2} = \Phi^2 y$; $\frac{dy}{dx}(0) = 0$; $y(1) = 1$. Solve this differential equation using the 'dsolve' command to obtain a closed-form solution. Plot the profile.

14. Use Maple to find the Laplace transforms of the following functions.

(1) sinh(at) + cosh(at)
(2) exp(at) sinh(at)
(3) $\frac{1}{1+t}$

15. Use Maple to find the inverse Laplace transforms of the following functions.

(1) $\frac{1}{s(s+1)} - \frac{1}{1+s^2}$
(2) $\frac{1}{1+s\,\exp(s)}$
(3) $\frac{1}{1+\sqrt{s}}$

References

1. Garvan, F.: The Maple Book, p. 479. Chapman & Hall/CRC, Boca Raton (2002)
2. Abel, M.L., Braselton, J.P.: Differential Equations with Maple V, 3rd edn. Academic Press, London (2001)
3. Davis, M.E.: Numerical Methods and Modeling for Chemical Engineers. John Wiley & Sons, Chichester (1984)
4. Rice, R.G., Do, D.D.: Applied Mathematics and Modeling for Chemical Engineers. John Wiley & Sons, Inc., Chichester (1995)